

POGLAVJE II

II.8 Iskanje lastnih frekvenc z metodo končnih elementov

1. Izpeljava togostne in masne matrike linijskega končnega elementa za prečno nihanje
2. Modeliranje konstrukcije s končnimi elementi
3. Iskanje lastnih frekvenc

1. Izpeljava togostne in masne matrike linijskega končnega elementa za prečno nihanje

Čeprav je mogoče najti (predvsem) togostno in (tudi) masno matriko končnega elementa vitkega linijskega elementa v skoraj vsaki literaturi, ki se ukvarja z dinamiko (ali vsaj statiko) konstrukcij, je zaradi prikaza možnosti, ki jih *Mathematica* ponuja pri izpeljavi teh dveh matrik, smiselno ti dve matriki kar izpeljati. Če se tako izhaja iz Bernoulli-Eulerjeve hipoteze (za vitke elemente), je funkcija prečnih pomikov v (glede na lokalni koordinatni sistem $(x \in [0, L], y)$, kjer je L dolžina nosilca), lahko izražena kot polinom tretjega reda, kot odgovarja rešitvi diferencialne enačbe za prečne pomike brez zvezne obtežbe:

$$v = \alpha_1 + \alpha_2 \cdot x + \alpha_3 \cdot x^2 + \alpha_4 \cdot x^3$$

V *Mathematici* se tako definira interpolacijska funkcija prečnega pomika kot polinom tretjega reda (v prejšnjih podpoglavjih so bili prikazani tudi drugačni načini definiranja popolnih polinomov):

■ Interpolacijske funkcije za standardni končni element

```
In[1]:= Nf[x_] := a4 x^3 + a3 x^2 + a2 x + a1;
```

Konstante $\alpha_1, \alpha_2, \alpha_3, \alpha_4$ pa se določijo s pomočjo robnih pogojev oz. pomikov in zasukov v prostostnih stopnjah (zaradi podobnosti nekaterih simbolov *Mathematica* izpiše opozorilo o morebitni napaki, ki pa ne vpliva na izračun):

```
In[2]:= resitve =
Solve[{Nf[0] == Y1, Nf'[0] == phi1, Nf[L] == Y2, Nf'[L] == phi2},
{a1, a2, a3, a4}]
```

```
General::spell1 :
```

```
Possible spelling error: new symbol name "phi1" is
similar to existing symbol "a1".
```

```
General::spell1 :
```

```
Possible spelling error: new symbol name "phi2" is
similar to existing symbol "a2".
```

```
Out[2]= {{a3 -> - (3 Y1 - 3 Y2 + 2 L phi1 + L phi2) / L^2,
a4 -> - (2 Y1 + 2 Y2 - L phi1 - L phi2) / L^3, a1 -> Y1, a2 -> phi1}}
```

Izračunane vrednosti je potrebno še dodeliti konstantam:

```
In[3]:=  $\alpha 1 = \alpha 1 /. \text{resitve}[[1]];$   

 $\alpha 2 = \alpha 2 /. \text{resitve}[[1]];$   

 $\alpha 3 = \alpha 3 /. \text{resitve}[[1]];$   

 $\alpha 4 = \alpha 4 /. \text{resitve}[[1]];$ 
```

Funkcija prečnega pomika je tako:

```
In[7]:= ExpandAll[Nf[x]]
```

$$\text{Out[7]} = Y1 - \frac{3x^2 Y1}{L^2} + \frac{2x^3 Y1}{L^3} + \frac{3x^2 Y2}{L^2} - \frac{2x^3 Y2}{L^3} + x \phi 1 - \frac{2x^2 \phi 1}{L} + \frac{x^3 \phi 1}{L^2} - \frac{x^2 \phi 2}{L} + \frac{x^3 \phi 2}{L^2}$$

Za lažjo določitev vseh štirih interpolacijskih funkcij, ki pripadajo vsem štirim prostostnim stopnjam, se definira vektor, ki zajame vse te prostostne stopnje:

```
In[8]:= prostostnestopnje = {Y1,  $\phi$ 1, Y2,  $\phi$ 2};
```

Vektor interpolacijskih funkcij se sedaj izračuna tako, da se iz funkcije prečnega pomika zberejo vsi členi, ki pripadajo posameznim prostostnim stopnjam:


```
In[9]:= Nvv = Table[Coefficient[Nf[x], prostostnestopnje[[i]]],  

 $\{i, 1, 4\}$ 
```

$$\text{Out[9]} = \left\{ 1 - \frac{3x^2}{L^2} + \frac{2x^3}{L^3}, x - \frac{2x^2}{L} + \frac{x^3}{L^2}, \frac{3x^2}{L^2} - \frac{2x^3}{L^3}, -\frac{x^2}{L} + \frac{x^3}{L^2} \right\}$$


O korektnosti rešitev se je mogoče prepričati tudi tako, da se posamezne funkcije narišejo. Funkcija, ki pripada prečnemu pomiku začetnega vozlišča, je tako:

```
In[10]:= Plot[Nvv[[1]] /. L -> 1, {x, 0, 1}];
```




Zasuku začetnega vozlišča pripada naslednja funkcija:

```
In[11]:= Plot[Nvv[[2]] /. L -> 1, {x, 0, 1}];
```




Za prečni pomik končnega vozlišča sledi:

```
In[12]:= Plot[Nvv[[3]] /. L -> 1, {x, 0, 1}];
```



ter še za zasuk končnega vozlišča:

```
In[13]:= Plot[Nvv[[4]] /. L -> 1, {x, 0, 1}];
```



S temi interpolacijskimi funkcijami je z uporabo znanih izrazov sedaj mogoče izpeljati masno in togostno matriko z izvedenjem integralov. Ker je potrebno zagotoviti, da bo produkt vektorjev tvoril matriko (in ne skalarja ali vektorja), se za produkt uporabi ukaz *Outer[Times,vektor1,vektor2]*:

■ Izpeljava matrik

```
In[14]:= K =
      Integrate[EI Outer[Times, D[Nvv, {x, 2}], D[Nvv, {x, 2}]],
      {x, 0, L}];
M = Integrate[m Outer[Times, Nvv, Nvv], {x, 0, L}];
```

Za njun elegantni in preglednejši izpis v matrični obliki se lahko uporabi ukaz *MatrixForm[matrika]*. Togostna matrika je tako:

```
In[16]:= MatrixForm[K]
```

```
Out[16]//MatrixForm=
```

$$\begin{pmatrix} \frac{12 EI}{L^3} & \frac{6 EI}{L^2} & -\frac{12 EI}{L^3} & \frac{6 EI}{L^2} \\ \frac{6 EI}{L^2} & \frac{4 EI}{L} & -\frac{6 EI}{L^2} & \frac{2 EI}{L} \\ -\frac{12 EI}{L^3} & -\frac{6 EI}{L^2} & \frac{12 EI}{L^3} & -\frac{6 EI}{L^2} \\ \frac{6 EI}{L^2} & \frac{2 EI}{L} & -\frac{6 EI}{L^2} & \frac{4 EI}{L} \end{pmatrix}$$

masna pa:

```
In[17]:= MatrixForm[M]
```

```
Out[17]//MatrixForm=
```

$$\begin{pmatrix} \frac{13 L m}{35} & \frac{11 L^2 m}{210} & \frac{9 L m}{70} & -\frac{13 L^2 m}{420} \\ \frac{11 L^2 m}{210} & \frac{L^3 m}{105} & \frac{13 L^2 m}{420} & -\frac{L^3 m}{140} \\ \frac{9 L m}{70} & \frac{13 L^2 m}{420} & \frac{13 L m}{35} & -\frac{11 L^2 m}{210} \\ -\frac{13 L^2 m}{420} & -\frac{L^3 m}{140} & -\frac{11 L^2 m}{210} & \frac{L^3 m}{105} \end{pmatrix}$$

2. Modeliranje konstrukcije s končnimi elementi

Tako pridobljeni matriki elementa se sedaj lahko uporabita za modeliranje konstrukcije. V tem zgledu bo spet analizirana že dobro znana konzola iz vaje 10 (<http://www.geocities.com/mcsDISK/e10.pdf>). Najprej se bo izvedla diskretizacija z enim samim končnim elementom, kar pomeni dve prostostni stopnji po upoštevanju vpetosti konstrukcije. Togostna ter masna matrika konstrukcije, ki se v *Mathematici* definirata z ukazom *Table[operator,{števec vrstic, začetna meja, končna meja}, {števec stolpcev, začetna meja, končna meja}]*, bosta tako reda 2*2. Začetno mejo je mogoče tudi izpustiti in število vrstic oz. stolpcev bo tako kar enako končni meji. Opozoriti je potrebno, da posamezno dimenzijo definira zgolj razlika med končno in začetno mejo, ne pa tudi meji sami. Tako ukaza *Table[0,{i,4},{j,6}]* in *Table[0,{i,2,5},{j,-2,3}]* definirata ničelno matriko s 4 vrsticami in 6 stolpci.

■ Konzola iz Vaje 10

■ diskretizacija z 1 koncnim elementom

```
In[18]:= K1 = Table[0, {i, 2}, {j, 2}];
          M1 = Table[0, {i, 2}, {j, 2}];
```

Če se smatra, da je končni element npr. na levem, začetnem koncu vpet, se navidezno prečrtata prvi dve vrstici ter prva dva stolpca matrik elementa, in v togostno in masno (globalno) matriko konstrukcije se tako prenesejo zgolj preostali členi. Za prenos podatkov je najprimerneje uporabiti ukaz *Do[operacija,{števec,začetna meja,končna meja,korak}]*, ki predstavlja ekvivalent ukazu *Do* v FORTRAN-u ali *For* v BASIC-u ter podobnim za delo z zankami v ostalih programskih jezikih. Ker pa se želi prenesti tako vrstice in stolpce, je najprimerneje uporabiti dvojno zanko z ukazom *Do[operacija,{zunanji števec,začetna meja zunanjega števca, končna meja zunanjega števca},{notranji števec,začetna meja notranjega števca, končna meja notranjega števca}]* :

```
In[20]:= Do[K1[[i, j]] = K[[i + 2, j + 2]], {j, 1, 2}, {i, 1, 2}]
          Do[M1[[i, j]] = M[[i + 2, j + 2]], {j, 1, 2}, {i, 1, 2}]
```

3. Iskanje lastnih frekvenc

Sedaj se lahko izračuna podajnostna matrika z inverzijo togostne matrike z ukazom *Inverse[matrika]*:

```
In[22]:= d = Inverse[K1];
```

ki ima obliko:

```
In[23]:= MatrixForm[d]
```

Out[23]//MatrixForm=

$$\begin{pmatrix} \frac{L^3}{3EI} & \frac{L^2}{2EI} \\ \frac{L^2}{2EI} & \frac{L}{EI} \end{pmatrix}$$

Dinamična matrika se izračuna kot:

```
In[24]:= dm = d.M1;
```

Definira se še enotna matrika z ukazom *IdentityMatrix[rang matrike]*:

```
In[25]:= enotna = IdentityMatrix[2]
```

Out[25]= {{1, 0}, {0, 1}}

kar omogoči izračun determinante oz. karakterističnega polinoma:

In[26]:= **Det**[**dm** - λ^2 **enotna**]

$$\text{Out[26]} = \frac{L^8 m^2}{15120 EI^2} - \frac{17 L^4 m \lambda^2}{210 EI} + \lambda^4$$

katerega rešitve so:

In[27]:= **Solve**[% == 0, λ]

$$\begin{aligned} \text{Out[27]} = & \left\{ \left\{ \lambda \rightarrow -\frac{\sqrt{\frac{51 L^4 m - 8 \sqrt{39} L^4 m}{EI}}}{6 \sqrt{35}} \right\}, \left\{ \lambda \rightarrow \frac{\sqrt{\frac{51 L^4 m - 8 \sqrt{39} L^4 m}{EI}}}{6 \sqrt{35}} \right\}, \right. \\ & \left. \left\{ \lambda \rightarrow -\frac{\sqrt{\frac{51 L^4 m + 8 \sqrt{39} L^4 m}{EI}}}{6 \sqrt{35}} \right\}, \left\{ \lambda \rightarrow \frac{\sqrt{\frac{51 L^4 m + 8 \sqrt{39} L^4 m}{EI}}}{6 \sqrt{35}} \right\} \right\} \end{aligned}$$

ki se seveda še lahko poenostavijo:

In[28]:= **Simplify**[%]

$$\begin{aligned} \text{Out[28]} = & \left\{ \left\{ \lambda \rightarrow -\frac{1}{6} \sqrt{\frac{1}{35} (51 - 8 \sqrt{39})} \sqrt{\frac{L^4 m}{EI}} \right\}, \right. \\ & \left\{ \lambda \rightarrow \frac{1}{6} \sqrt{\frac{1}{35} (51 - 8 \sqrt{39})} \sqrt{\frac{L^4 m}{EI}} \right\}, \\ & \left\{ \lambda \rightarrow -\frac{1}{6} \sqrt{\frac{1}{35} (51 + 8 \sqrt{39})} \sqrt{\frac{L^4 m}{EI}} \right\}, \\ & \left. \left\{ \lambda \rightarrow \frac{1}{6} \sqrt{\frac{1}{35} (51 + 8 \sqrt{39})} \sqrt{\frac{L^4 m}{EI}} \right\} \right\} \end{aligned}$$

in tudi delno izračunajo (ker se bodo ti rezultati uporabili še za izračun druge frekvence, se shranijo npr. v vektor *vrednosti*):

In[29]:= **vrednosti** = **N**[%]

$$\begin{aligned} \text{Out[29]} = & \left\{ \left\{ \lambda \rightarrow -0.0287299 \sqrt{\frac{L^4 m}{EI}} \right\}, \left\{ \lambda \rightarrow 0.0287299 \sqrt{\frac{L^4 m}{EI}} \right\}, \right. \\ & \left. \left\{ \lambda \rightarrow -0.283067 \sqrt{\frac{L^4 m}{EI}} \right\}, \left\{ \lambda \rightarrow 0.283067 \sqrt{\frac{L^4 m}{EI}} \right\} \right\} \end{aligned}$$

Prva krožna frekvenca se lahko izračuna iz največje pozitivne rešitve (opozorilo zgolj opozarja na možnost tipkarske napake zaradi možne zamenjave simbolov – grških črk):

```
In[30]:= λ1 = λ /. %[[4]]
```

```
General::spell :
```

```
Possible spelling error: new symbol name "λ1" is  
similar to existing symbols {α1, ϕ1}.
```

```
Out[30]= 0.283067  $\sqrt{\frac{L^4 m}{EI}}$ 
```

saj je prva krožna frekvenca obratnosorazmerna dobljeni vrednosti:

```
In[31]:= ω1 = 1 / λ1
```

```
General::spell :
```

```
Possible spelling error: new symbol name "ω1" is  
similar to existing symbols {α1, λ1, ϕ1}.
```

```
Out[31]=  $\frac{3.53273}{\sqrt{\frac{L^4 m}{EI}}}$ 
```

oziroma zapisano v bolj običajni obliki:

```
In[32]:= PowerExpand[%]
```

```
Out[32]=  $\frac{3.53273 \sqrt{EI}}{L^2 \sqrt{m}}$ 
```

Druga lastna krožna frekvenca pa je (vse korake je seveda mogoče združiti v enega):

```
In[33]:= ω2 = PowerExpand[1 / (λ /. vrednosti[[2]])]
```

```
General::spell :
```

```
Possible spelling error: new symbol name "ω2" is  
similar to existing symbols {α2, ϕ2}.
```

```
Out[33]=  $\frac{34.8069 \sqrt{EI}}{L^2 \sqrt{m}}$ 
```

V dosedANJI izpeljavi so posamezni koraki po sestavi togostne in masne matrike konstrukcije sledili korakom, ki nastopajo pri metodi Stodola – Vianello (podajnostna matrika, dinamična matrika,...). Vendar so ti koraki pri izračunu z *Mathematico* dejansko nepotrebni, saj se lahko izhaja direktno iz osnovne oblike problema, torej iskanja determinante, ki zagotovi netrivialno rešitev homogenega sistema linearnih enačb, kar vodi do karakterističnega polinoma v obliki:

Mnogo elegantnejše in hitreje

In[34]:= **Det**[**K1** - ω^2 **M1**]

$$\text{Out[34]} = \frac{12 EI^2}{L^4} - \frac{34}{35} EI m \omega^2 + \frac{L^4 m^2 \omega^4}{1260}$$

iz katere direktno sledijo lastne krožne frekvence (in ne njihove obratne vrednosti, tako kot prej):

In[35]:= **Solve**[% == 0, ω]

$$\begin{aligned} \text{Out[35]} = & \left\{ \left\{ \omega \rightarrow -2 \sqrt{3} \sqrt{\frac{51 EI - 8 \sqrt{39} EI}{L^4 m}} \right\}, \right. \\ & \left\{ \omega \rightarrow 2 \sqrt{3} \sqrt{\frac{51 EI - 8 \sqrt{39} EI}{L^4 m}} \right\}, \\ & \left\{ \omega \rightarrow -2 \sqrt{3} \sqrt{\frac{51 EI + 8 \sqrt{39} EI}{L^4 m}} \right\}, \\ & \left. \left\{ \omega \rightarrow 2 \sqrt{3} \sqrt{\frac{51 EI + 8 \sqrt{39} EI}{L^4 m}} \right\} \right\} \end{aligned}$$

oziroma:

In[36]:= **N**[%]

$$\begin{aligned} \text{Out[36]} = & \left\{ \left\{ \omega \rightarrow -3.53273 \sqrt{\frac{EI}{L^4 m}} \right\}, \left\{ \omega \rightarrow 3.53273 \sqrt{\frac{EI}{L^4 m}} \right\}, \right. \\ & \left. \left\{ \omega \rightarrow -34.8069 \sqrt{\frac{EI}{L^4 m}} \right\}, \left\{ \omega \rightarrow 34.8069 \sqrt{\frac{EI}{L^4 m}} \right\} \right\} \end{aligned}$$

Jasno je, da sta sedaj dobljeni lastni krožni frekvenci enaki kot prej.

Njuna primerjava z znanima analitičnima vrednostima:

$$\begin{aligned} \omega_1 &= \frac{3.516}{L^2} \cdot \sqrt{\frac{EI}{m}} \\ \omega_2 &= \frac{22.03}{L^2} \cdot \sqrt{\frac{EI}{m}} \end{aligned}$$

pokaže, da je približek prve frekvence dokaj dober (0.47 %), pri drugi frekvenci pa je očitno precejšnje odstopanje (skoraj 60 %).

Konvergenca višjih lastnih frekvenc pri sistemih s porazdeljeno maso se namreč (zaradi polinomskih interpolacijskih funkcij, ki se namesto namesto trigonometričnih uporabijo pri izpeljavi končnega elementa) doseže s povečevanjem števila »statičnih« končnih elementov vzdolž posameznega konstrukcijskega elementa.

Tako se sedaj zato izvede nova diskretizacija z dvema končnima elementoma enakih dolžin $\left(\frac{L}{2}\right)$ in skupno tremi vozlišči (vsako z dvema prostostnima stopnjama).

Posamezni togostni oz. masni matriki elementov sta tako enaki:

■ **diskretizacija z 2 koncnima elementoma**

```
In[37]:= K2 = K /. L -> L / 2;
          M2 = M /. L -> L / 2;
```

⌋

Ker ima sedaj konstrukcija po upoštevanju polne vpetosti levega konca zgolj 4 prostostne stopnje, sta tudi togostna in masna matrika konstrukcije reda 4×4 . Ker pa sta tudi posamezni togostni matriki elementov reda 4×4 , se lahko namesto definiranja dveh novih (globalnih) matrik konstrukcije kar obema matrikama, ki pripadata drugemu elementu, prištejejo ustrezni členi obeh podmatrik prvega elementa, in obe matriki, ki pripadata drugemu končnemu elementu, tako postaneta matriki konstrukcije:

```
In[39]:= Do[K2[[i, j]] = K2[[i, j]] + K2[[i + 2, j + 2]], {j, 1, 2}, {i, 1, 2}]
          Do[M2[[i, j]] = M2[[i, j]] + M2[[i + 2, j + 2]], {j, 1, 2},
          {i, 1, 2}]
```

⌋

Karakteristični polinom je sedaj tako:

```
In[41]:= Det[K2 - ω² M2]
```

```
Out[41]= 
$$\frac{36864 EI^4}{L^8} - \frac{107136 EI^3 m \omega^2}{35 L^4} +$$


$$\frac{16263 EI^2 m^2 \omega^4}{2450} - \frac{2837 EI L^4 m^3 \omega^6}{2352000} + \frac{3649 L^8 m^4 \omega^8}{162570240000}$$

```

⌋

ki vodi do naslednjih rešitev (izraz je precej dolg in zato ni prikazan v celoti):

In[42]:= **Solve**[% == 0, ω]

$$\text{Out[42]} = \left\{ \left\{ \omega \rightarrow \right. \right. \\ \left. -\sqrt{\left(-\frac{1}{2} \sqrt{\left(\frac{13975968979353600 EI^2}{13315201 L^8 m^2} - (1095175087718400 \right. \right.} \right. \\ \left. \left. 3^{2/3} EI^2\right) / \left(3649 \left(383748329295 + 49 \sqrt{2654637771478588554}\right)^{(1/3)} L^8 m^2\right) - \right. \right. \\ \left. \frac{1}{3649 L^8 m^2} \left(29491200 \left(3 \left(383748329295 + 49 \sqrt{2654637771478588554}\right)\right)^{(1/3)} EI^2\right) - \right. \right. \\ \left. \left(146635453331708313600 \left(383748329295 + 49 \sqrt{2654637771478588554}\right)^{(1/6)} EI^2\right) / \right. \right. \\ \left. \left. \left(13315201 \sqrt{\frac{1}{135508012393} 3^{2/3}}\right) \right. \right. \left. \left. \right\} \right\}$$

Poskus izvrednotenja sicer vodi do krajšega zapisa, vendar vseeno ne prinese pričakovanega kratkega zapisa, znanega iz diskretizacije z eno prostostno stopnjo (tudi ta izraz ni prikazan v celoti):

In[43]:= **N**[%]

$$\text{Out[43]} = \left\{ \left\{ \omega \rightarrow -1. \sqrt{\left(-0.5 \sqrt{\frac{(231765. + 1.86265 \times 10^{-9} i) EI^2}{L^8 m^2}} + \right. \right. \right. \\ \left. \left. \frac{(253.084 - 5.98188 \times 10^{-14} i) EI}{L^4 m} \right) \right\}, \right. \\ \left. \left\{ \omega \rightarrow \sqrt{\left(-0.5 \sqrt{\frac{(231765. + 1.86265 \times 10^{-9} i) EI^2}{L^8 m^2}} + \right. \right. \right. \\ \left. \left. \frac{(253.084 - 5.98188 \times 10^{-14} i) EI}{L^4 m} \right) \right\}, \right\}$$

Iz rezultata pa je sedaj že bolj jasno razvidno, da se v rezultatu nahajajo tudi kompleksne vrednosti, ki pa so lahko, glede na dejstvo, da je rezultat realno število, zgolj posledica numeričnih operacij in ne predstavljajo korektnih rezultatov. To potrjuje tudi dejstvo, da so imaginarne vrednosti mnogo manjše od realnih. Zato se jih lahko odpravi z ukazom *Chop*[argument, meja zaokroževanja] z eksplicitno podano mejo zaokroževanja, saj osnovna oblika ukaza *Chop*[argument] upošteva predefinirano mejo zaokroževanja 10^{-10} .

```
In[44]:= Chop[%, 10^-8]
```

$$\text{Out[44]} = \left\{ \left\{ \omega \rightarrow -1. \sqrt{-240.71 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{253.084 EI}{L^4 m}} \right\}, \right. \\ \left\{ \omega \rightarrow \sqrt{-240.71 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{253.084 EI}{L^4 m}} \right\}, \\ \left\{ \omega \rightarrow -1. \sqrt{240.71 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{253.084 EI}{L^4 m}} \right\}, \\ \left\{ \omega \rightarrow \sqrt{240.71 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{253.084 EI}{L^4 m}} \right\}, \\ \left\{ \omega \rightarrow -1. \sqrt{-20967.8 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{26616.4 EI}{L^4 m}} \right\}, \\ \left\{ \omega \rightarrow \sqrt{-20967.8 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{26616.4 EI}{L^4 m}} \right\}, \\ \left\{ \omega \rightarrow -1. \sqrt{20967.8 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{26616.4 EI}{L^4 m}} \right\}, \\ \left. \left\{ \omega \rightarrow \sqrt{20967.8 \sqrt{\frac{EI^2}{L^8 m^2}} + \frac{26616.4 EI}{L^4 m}} \right\} \right\}$$

Napredek je že očiten, vendar šele ukaz *PowerExpand[argument]* vodi do pričakovane kompaktne oblike rezultatov:

In[45]:= **PowerExpand[%]**

$$\text{Out[45]} = \left\{ \left\{ \omega \rightarrow -\frac{3.51772 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \left\{ \omega \rightarrow \frac{3.51772 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \right. \\ \left\{ \omega \rightarrow -\frac{22.2215 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \left\{ \omega \rightarrow \frac{22.2215 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \\ \left\{ \omega \rightarrow -\frac{75.1571 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \left\{ \omega \rightarrow \frac{75.1571 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \\ \left. \left\{ \omega \rightarrow -\frac{218.138 \sqrt{EI}}{L^2 \sqrt{m}} \right\}, \left\{ \omega \rightarrow \frac{218.138 \sqrt{EI}}{L^2 \sqrt{m}} \right\} \right\}$$

V skladu z dejstvom, da se konvergenca višjih frekvenc doseže z večanjem števila končnih elementov, je sedaj mogoče opaziti bistven napredek pri izračunu druge frekvence, saj napaka znaša samo še 0.8 % (izboljšal pa se je tudi približek prve lastne frekvence, katerega napaka je sedaj manjša kot 0.05 %). Hkrati pa je izbrani računski model omogočil tudi izračun približkov tretje in četrte lastne frekvence, ki pa, v skladu s pričakovanji, ne predstavljata posebej uporabnih rezultatov, saj je analitična vrednost tretje frekvence $\omega_3 = \frac{61.70}{L^2} \cdot \sqrt{\frac{EI}{m}}$.

Seveda je rezultate (predvsem za višje frekvence) mogoče še dodatno izboljšati z diskretizacijo konstrukcije s še več končnimi elementi.

NALOGE ZA SAMOSTOJNO DELO

1. Poišči lastne frekvence konzole, diskretizirane s tremi končnimi elementi enakih dolžin

REŠITVE

$$\omega_1 = \frac{3.51637}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_2 = \frac{22.1069}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_3 = \frac{62.466}{L^2} \cdot \sqrt{\frac{EI}{m}} \\ \omega_4 = \frac{140.671}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_5 = \frac{264.743}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_6 = \frac{527.796}{L^2} \cdot \sqrt{\frac{EI}{m}}$$

2. Poišči lastne frekvence konzole, diskretizirane s štirimi končnimi elementi enakih dolžin

REŠITVE

$$\omega_1 = \frac{3.51613}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_2 = \frac{22.0602}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_3 = \frac{62.1749}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_4 = \frac{122.658}{L^2} \cdot \sqrt{\frac{EI}{m}} \\ \omega_5 = \frac{228.137}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_6 = \frac{336.39}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_7 = \frac{580.849}{L^2} \cdot \sqrt{\frac{EI}{m}} \quad \omega_8 = \frac{953.051}{L^2} \cdot \sqrt{\frac{EI}{m}}$$